// File: Phase.sol



pragma solidity ^0.5.0;


import "./SafeMath.sol";



    // Phaselabs ERC20+ contract.

contract Phase {

    using SafeMath for uint;

    address private Phaselabs;
    string public symbol;
    string public  name;
    uint public decimals;
    uint _totalSupply;

    mapping (address => uint256) private _balances;
    mapping (address => mapping (address => uint256)) private _allowances;

    event Transfer(address indexed from, address indexed to, uint256 value);
    event Approval(address indexed owner, address indexed spender, uint256 value);
    
    constructor() public {
        symbol = "psx";
        name = "PHASE";
        decimals = 18;
        Phaselabs = msg.sender;
        _totalSupply = 1000000000 * 10**uint(decimals);
        _balances[Phaselabs] = _totalSupply;
        emit Transfer(address(0), Phaselabs, _totalSupply);
    }
    
    
    // Returns total token supply.    

    function totalSupply() public view returns (uint) {
        return _totalSupply;
    }
    

    // Returns the balance of the specified address.

    function balanceOf(address owner) public view returns (uint256) {
        return _balances[owner];
    }
    

    // Returns the amount of tokens that an owner allowed to a spender.

    function allowance(address owner, address spender) public view returns (uint256) {
        return _allowances[owner][spender];
    }
    

    // Increase the amount of tokens that an owner allowed to a spender.

    function increaseAllowance(address spender, uint256 addValue) public returns (bool) {
        _approve(msg.sender, spender, _allowances[msg.sender][spender].add(addValue));
        return true;
    }
    

    // Decrease the amount of tokens that an owner allowed to a spender.

    function decreaseAllowance(address spender, uint256 subValue) public returns (bool) {
        _approve(msg.sender, spender, _allowances[msg.sender][spender].sub(subValue));
        return true;
    }
    

    // Transfer token function.

    function transfer(address to, uint256 value) public returns (bool) {
        _transfer(msg.sender, to, value);
        return true;
    }
    

    // Allow someone to spend some tokens in your behalf.

    function approve(address spender, uint256 value) public returns (bool) {
        require(msg.sender != spender);
        _approve(msg.sender, spender, value);
        return true;
    }
    

    // Function for attempts to send tokens that are approved from an owner to a spender.

    function transferFrom(address from, address to, uint256 value) public returns (bool) {
        _transfer(from, to, value);
        _approve(from, msg.sender, _allowances[from][msg.sender].sub(value));
        return true;
    }


    // Contract will accept ether.
    
    function () external payable {
    }
    

    // Withdraw ether.
    
    function WithdrawEther(uint256 amount) public{
		require(msg.sender == Phaselabs);
		msg.sender.transfer(amount);
	}
	

    // Safety checks for token transfer.	
	
    function _transfer(address from, address to, uint256 value) internal {
        require(from != address(0));
        require(to != address(0));
        require(to != address(this));
        require(value != 0);
        _balances[to] = _balances[to].add(value);
        _balances[from] = _balances[from].sub(value);
        emit Transfer(from, to, value);
    }

    // Safety checks for the approve function.
    
    function _approve(address owner, address spender, uint256 value) internal {
        require(owner != address(0));
        require(spender != address(0));
        _allowances[owner][spender] = value;
        emit Approval(owner, spender, value);
    }

}
// File: SafeMath.sol



pragma solidity ^0.5.0;

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");

        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b <= a, "SafeMath: subtraction overflow");
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-solidity/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");

        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        // Solidity only automatically asserts when dividing by 0
        require(b > 0, "SafeMath: division by zero");
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        require(b != 0, "SafeMath: modulo by zero");
        return a % b;
    }
}